
1 Corrections required

1.1 fl_NBIOT_Get_EutraCarrierFreq
	Template name
	fl_NBIOT_GetEutraCarrierFreq

	Reason for change
	In the current TTCN implementation of 22.4.12, f_NBIOT_Init-> fl_NBIOT_Common_Init-> f_NBIOT_CellArray_Init-> f_NBIOT_CellInfo_Init-> v_CellInfo := cs_NBIOT_CellInfoDef(v_CellIds,
                                       v_SysinfoCombination,
                                       p_FrequencyBand,
                                       v_UL_DL_Earfcn,
                                       v_NAS_Param,
                                       v_BCCH_Info,
                                       v_CellTiming,
                                       p_SubCarrierSpacingUL,
                                       fl_NBIOT_Get_EUTRACarrierFreq(p_CellId)); is called. 
This is called to initialise Ncell 10, which is on a different band than Ncell 1.
However, the function fl_NBIOT_Get_EutraCarrierFreq assigned v_FreqLMH based on px_ePrimaryFrequencyBand and px_ePrimaryBandChannelBandwidth. As a result, for the GuardBand mode of operation, or the InBand_SamePCI mode of operation, the LTE cell EARFCN is set to that of Ncell 1, not Ncell 10. This results in incorrect EARFCN for Ncell 10 as well and as a result the UE never finds Ncell 10.
This needs to be corrected.                                               


	Summary of change
	Added a argument to function fl_NBIOT_Get_EUTRACarrierFreq, a boolean flag p_IsConfiguredBandPrimary, and set it to true by default. And based on this flag, assigned v_FreqLMH. 

Please see below.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>
function fl_NBIOT_Get_EUTRACarrierFreq(NBIOT_CellId_Type    p_CellId,
                                      EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN)
    return integer
  {
    var CellCarrierFreqEUTRA_Type v_CarrierFreq;
    var template (value) NBIOT_EUTRAFrequency_fList_Type v_Freq := cs_NBIOT_EUTRAFrequencyInitDummy;   //Initialise with Dummy value
    var NBIOT_EUTRAFrequency_fList_Type v_EUTRA_Freq := valueof(v_Freq);
    var FrequencyLowMidHigh_Type v_FreqLMH := f_EUTRA_InitFrequencyLowMidHigh(px_ePrimaryFrequencyBand,
                                                      px_ePrimaryBandChannelBandwidth,
                                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth));
    // Most of the bands assigns f1->Mid  f2->High  f3->Low
    v_EUTRA_Freq.f1 := v_FreqLMH.FrequencyMid;
    v_EUTRA_Freq.f2 := v_FreqLMH.FrequencyHigh;
    v_EUTRA_Freq.f3 := v_FreqLMH.FrequencyLow;
    select (p_CellId) {
      case (nbiot_Cell1, nbiot_Cell2, nbiot_Cell4, nbiot_Cell5, nbiot_Cell11, nbiot_Cell18) { // @sic R5-176139 sic@
        // For cell frequency of f1:  v_Frequency_PrimaryBand.f1
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
        case (nbiot_Cell3, nbiot_Cell12, nbiot_Cell23) { // @sic R5s170476 sic@
        // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
        v_CarrierFreq := v_EUTRA_Freq.f2;
      }
      case (nbiot_Cell6, nbiot_Cell13) { // @sic R5s170476 sic@
        // For cell frequency of f3:         v_Frequency_PrimaryBand.f3
        v_CarrierFreq := v_EUTRA_Freq.f3;
      }
      case (nbiot_Cell10) {
        // For cell frequency of f5:         v_Frequency_SecondaryBand.f1 (f5)
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
      case (nbiot_Cell50, nbiot_Cell51, nbiot_Cell53, nbiot_Cell62, nbiot_Cell63) { // @sic R5s170476 sic@
        // For cell frequency of f1:         v_Frequency_PrimaryBand.f1
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
      case (nbiot_Cell52, nbiot_Cell61) {
        if (p_CellsOnPLMN == LTE_AllCellsOnSamePLMN) { // @sic R5s170476 sic@
          // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
          v_CarrierFreq := v_EUTRA_Freq.f2;
        } else { // different PLMN
          // For cell frequency of f1:         v_Frequency_PrimaryBand.f1
          v_CarrierFreq := v_EUTRA_Freq.f1;
        }
      }
      case (nbiot_Cell54, nbiot_Cell55, nbiot_Cell56) { // @sic R5s170476 sic@
        // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
        v_CarrierFreq := v_EUTRA_Freq.f2;
      }
      case (nbiot_Cell57, nbiot_Cell59, nbiot_Cell60) {  // @sic R5s170476 sic@
        // For cell frequency of f3:         v_Frequency_PrimaryBand.f3
        v_CarrierFreq := v_EUTRA_Freq.f3;
      }
      case else {
        FatalError (__FILE__, __LINE__, "invalid value Cell Id for Carrier Freq");
      }
    }
    return v_CarrierFreq.dl_CarrierFreq_r9;
  }

<<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>
function fl_NBIOT_Get_EUTRACarrierFreq(NBIOT_CellId_Type    p_CellId,
                                      EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN,
                                      boolean p_IsConfiguredBandPrimary := true /*WA#GB*/)
    return integer
  {
    var CellCarrierFreqEUTRA_Type v_CarrierFreq;
    var template (value) NBIOT_EUTRAFrequency_fList_Type v_Freq := cs_NBIOT_EUTRAFrequencyInitDummy;   //Initialise with Dummy value
    var NBIOT_EUTRAFrequency_fList_Type v_EUTRA_Freq := valueof(v_Freq);
    var FrequencyLowMidHigh_Type v_FreqLMH := f_EUTRA_InitFrequencyLowMidHigh(px_ePrimaryFrequencyBand,
                                                      px_ePrimaryBandChannelBandwidth,
                                                      f_ConvertDL_BandwidthToUL(px_ePrimaryBandChannelBandwidth));
    if (p_IsConfiguredBandPrimary == false){
    v_FreqLMH := f_EUTRA_InitFrequencyLowMidHigh(px_eSecondaryFrequencyBand,
                                                      px_eSecondaryBandChannelBandwidth,
                                                      f_ConvertDL_BandwidthToUL(px_eSecondaryBandChannelBandwidth));
    }
    // Most of the bands assigns f1->Mid  f2->High  f3->Low
    v_EUTRA_Freq.f1 := v_FreqLMH.FrequencyMid;
    v_EUTRA_Freq.f2 := v_FreqLMH.FrequencyHigh;
    v_EUTRA_Freq.f3 := v_FreqLMH.FrequencyLow;
    select (p_CellId) {
      case (nbiot_Cell1, nbiot_Cell2, nbiot_Cell4, nbiot_Cell5, nbiot_Cell11, nbiot_Cell18) { // @sic R5-176139 sic@
        // For cell frequency of f1:  v_Frequency_PrimaryBand.f1
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
        case (nbiot_Cell3, nbiot_Cell12, nbiot_Cell23) { // @sic R5s170476 sic@
        // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
        v_CarrierFreq := v_EUTRA_Freq.f2;
      }
      case (nbiot_Cell6, nbiot_Cell13) { // @sic R5s170476 sic@
        // For cell frequency of f3:         v_Frequency_PrimaryBand.f3
        v_CarrierFreq := v_EUTRA_Freq.f3;
      }
      case (nbiot_Cell10) {
        // For cell frequency of f5:         v_Frequency_SecondaryBand.f1 (f5)
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
      case (nbiot_Cell50, nbiot_Cell51, nbiot_Cell53, nbiot_Cell62, nbiot_Cell63) { // @sic R5s170476 sic@
        // For cell frequency of f1:         v_Frequency_PrimaryBand.f1
        v_CarrierFreq := v_EUTRA_Freq.f1;
      }
      case (nbiot_Cell52, nbiot_Cell61) {
        if (p_CellsOnPLMN == LTE_AllCellsOnSamePLMN) { // @sic R5s170476 sic@
          // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
          v_CarrierFreq := v_EUTRA_Freq.f2;
        } else { // different PLMN
          // For cell frequency of f1:         v_Frequency_PrimaryBand.f1
          v_CarrierFreq := v_EUTRA_Freq.f1;
        }
      }
      case (nbiot_Cell54, nbiot_Cell55, nbiot_Cell56) { // @sic R5s170476 sic@
        // For cell frequency of f2:         v_Frequency_PrimaryBand.f2
        v_CarrierFreq := v_EUTRA_Freq.f2;
      }
      case (nbiot_Cell57, nbiot_Cell59, nbiot_Cell60) {  // @sic R5s170476 sic@
        // For cell frequency of f3:         v_Frequency_PrimaryBand.f3
        v_CarrierFreq := v_EUTRA_Freq.f3;
      }
      case else {
        FatalError (__FILE__, __LINE__, "invalid value Cell Id for Carrier Freq");
      }
    }
    return v_CarrierFreq.dl_CarrierFreq_r9;
  }

<<SKIPPED CODE>>


1.2 f_NBIOT_CellInfo_Init
	Template name
	f_NBIOT_CellInfo_Init

	Reason for change
	As a result of the change 1.1, the function f_NBIOT_CellInfo_Init also needs to be parameterized so that the boolean flag p_ IsConfiguredBandPrimary  can be passed further. 

	Summary of change
	Added an argument (a boolean flag p_ IsConfiguredBandPrimary) to function f_NBIOT_CellInfo_Init, and set its default to true.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	


Before Change:

	function f_NBIOT_CellInfo_Init(NBIOT_CellId_Type  p_CellId,
                                 integer  p_FrequencyBand,
                                 NB_Combination_Type p_SysinfoCombination,
                                 boolean p_DoAttachWithoutPDN,
                                 NB_SubCarrierSpacingUL_Type p_SubCarrierSpacingUL) return NBIOT_CellInfo_Type
  {
    var NB_CellIdentities_Type v_CellIds  := fl_NBIOT_InitCellIdentities(p_CellId);
    var CellTimingInfo_Type v_CellTiming := fl_NBIOT_InitialiseCellTiming(p_CellId);
    var NB_Combination_Type v_SysinfoCombination := fl_NBIOT_CheckSysinfoCombination(p_CellId, p_SysinfoCombination);
    var NB_CarrierFreq_Type v_UL_DL_Earfcn;
    var NAS_Parameter_Type v_NAS_Param;
    var template (value) NB_BcchInfo_Type  v_BCCH_Info;
    var template (value) NBIOT_CellInfo_Type v_CellInfo;
    var NB_OperationMode_Type v_OperationMode := f_NBIOT_OperationMode();
    // Initialise Frequencies f1/f2/f3/f5
    var NBIOT_Frequency_fList_Type v_Freq := f_NBIOT_InitFrequency_f1Tof3 (p_FrequencyBand, v_OperationMode); // @sic R5-171561, R5s170339 sic@
    // Initialised Frequency per cell. See 36.508 Table 8.1.4.2-1
    // Cell 1/2/4/11 -> Frequency f1
    // Cell 3/12/23 -> Frequency f2
    // Cell 6/13 -> Frequency f3
    // Cell 10 -> Frequency f5
    select (p_CellId) {
      case (nbiot_Cell1,
            nbiot_Cell2,
            nbiot_Cell4,
            nbiot_Cell5,
            nbiot_Cell10,
            nbiot_Cell11,
            nbiot_Cell18) { // @sic R5-176139 sic@
        v_UL_DL_Earfcn := v_Freq.f1;   //f1
      }
      case (nbiot_Cell3,
            nbiot_Cell12,
            nbiot_Cell23) {
        v_UL_DL_Earfcn := v_Freq.f2;  // f2
      }
      case (nbiot_Cell6, nbiot_Cell13) {
        v_UL_DL_Earfcn := v_Freq.f3;   // f3
      }
      case (nbiot_Cell14) {
        FatalError (__FILE__, __LINE__, "invalid cell Identity"); // @sic R5-171561 sic@
      }
      case else {
        FatalError (__FILE__, __LINE__, "invalid cell Identity");
      }
    }
    // Initialise NAS parameters
    v_NAS_Param := fl_NBIOT_InitNAS_forRRC_Cells(p_CellId);
    // Initialise SYSINFO
    v_BCCH_Info := f_NBIOT_InitSystemInformation(p_CellId,
                                                 v_CellIds.CellIdentity,
                                                 v_SysinfoCombination,
                                                 p_FrequencyBand,
                                                 v_OperationMode,
                                                 { f_NBIOT_PLMN_Identity(v_NAS_Param.Guti_Parameters.PLMN_Identity, p_DoAttachWithoutPDN) },
                                                 v_NAS_Param.TrackingAreaCode);
    v_CellInfo := cs_NBIOT_CellInfoDef(v_CellIds,
                                       v_SysinfoCombination,
                                       p_FrequencyBand,
                                       v_UL_DL_Earfcn,
                                       v_NAS_Param,
                                       v_BCCH_Info,
                                       v_CellTiming,
                                       p_SubCarrierSpacingUL,
                                       fl_NBIOT_Get_EUTRACarrierFreq(p_CellId)); // @sic R5w180205r1 sic@
    return valueof(v_CellInfo);
  }


After Change:

	function f_NBIOT_CellInfo_Init(NBIOT_CellId_Type  p_CellId,
                                 integer  p_FrequencyBand,
                                 NB_Combination_Type p_SysinfoCombination,
                                 boolean p_DoAttachWithoutPDN,
                                 NB_SubCarrierSpacingUL_Type p_SubCarrierSpacingUL,
                                 boolean p_IsConfiguredBandPrimary := true /*WA#GB*/) return NBIOT_CellInfo_Type
  {
    var NB_CellIdentities_Type v_CellIds  := fl_NBIOT_InitCellIdentities(p_CellId);
    var CellTimingInfo_Type v_CellTiming := fl_NBIOT_InitialiseCellTiming(p_CellId);
    var NB_Combination_Type v_SysinfoCombination := fl_NBIOT_CheckSysinfoCombination(p_CellId, p_SysinfoCombination);
    var NB_CarrierFreq_Type v_UL_DL_Earfcn;
    var NAS_Parameter_Type v_NAS_Param;
    var template (value) NB_BcchInfo_Type  v_BCCH_Info;
    var template (value) NBIOT_CellInfo_Type v_CellInfo;
    var NB_OperationMode_Type v_OperationMode := f_NBIOT_OperationMode();
    // Initialise Frequencies f1/f2/f3/f5
    var NBIOT_Frequency_fList_Type v_Freq := f_NBIOT_InitFrequency_f1Tof3 (p_FrequencyBand, v_OperationMode); // @sic R5-171561, R5s170339 sic@
    // Initialised Frequency per cell. See 36.508 Table 8.1.4.2-1
    // Cell 1/2/4/11 -> Frequency f1
    // Cell 3/12/23 -> Frequency f2
    // Cell 6/13 -> Frequency f3
    // Cell 10 -> Frequency f5
    select (p_CellId) {
      case (nbiot_Cell1,
            nbiot_Cell2,
            nbiot_Cell4,
            nbiot_Cell5,
            nbiot_Cell10,
            nbiot_Cell11,
            nbiot_Cell18) { // @sic R5-176139 sic@
        v_UL_DL_Earfcn := v_Freq.f1;   //f1
      }
      case (nbiot_Cell3,
            nbiot_Cell12,
            nbiot_Cell23) {
        v_UL_DL_Earfcn := v_Freq.f2;  // f2
      }
      case (nbiot_Cell6, nbiot_Cell13) {
        v_UL_DL_Earfcn := v_Freq.f3;   // f3
      }
      case (nbiot_Cell14) {
        FatalError (__FILE__, __LINE__, "invalid cell Identity"); // @sic R5-171561 sic@
      }
      case else {
        FatalError (__FILE__, __LINE__, "invalid cell Identity");
      }
    }
    // Initialise NAS parameters
    v_NAS_Param := fl_NBIOT_InitNAS_forRRC_Cells(p_CellId);
    // Initialise SYSINFO
    v_BCCH_Info := f_NBIOT_InitSystemInformation(p_CellId,
                                                 v_CellIds.CellIdentity,
                                                 v_SysinfoCombination,
                                                 p_FrequencyBand,
                                                 v_OperationMode,
                                                 { f_NBIOT_PLMN_Identity(v_NAS_Param.Guti_Parameters.PLMN_Identity, p_DoAttachWithoutPDN) },
                                                 v_NAS_Param.TrackingAreaCode);
    v_CellInfo := cs_NBIOT_CellInfoDef(v_CellIds,
                                       v_SysinfoCombination,
                                       p_FrequencyBand,
                                       v_UL_DL_Earfcn,
                                       v_NAS_Param,
                                       v_BCCH_Info,
                                       v_CellTiming,
                                       p_SubCarrierSpacingUL,
                                       fl_NBIOT_Get_EUTRACarrierFreq(p_CellId, -, p_IsConfiguredBandPrimary /*WA#GB*/)); // @sic R5w180205r1 sic@
    return valueof(v_CellInfo);
  }


1.3 f_NBIOT_CellArray_Init
	Template name
	f_NBIOT_CellArray_Init

	Reason for change
	As a result of the changes 1.1 and 1.2 above, in the function f_NBIOT_CellArray_Init, when the initialisation of Ncell 10 happens by calling function f_NBIOT_CellInfo_Init, the parameter p_IsConfiguredBandPrimary needs to be set to false.

	Summary of change
	Passed the value false for  parameter p_IsConfiguredBandPrimary when the function f_NBIOT_CellArray_Init is called to initialise the Ncell 10.
Please see below.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>

v_Index := f_NBIOT_CellInfo_GetIndex(nbiot_Cell10);
    v_NBIOT_Cells[v_Index] := f_NBIOT_CellInfo_Init(nbiot_Cell10,
                                                    v_SecondaryFrequencyBand,
                                                    p_SysinfoCombination,
                                                    p_DoAttachWithoutPDN,
                                                    v_SubCarrierSpacingUL);

 <<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>

v_Index := f_NBIOT_CellInfo_GetIndex(nbiot_Cell10);
    v_NBIOT_Cells[v_Index] := f_NBIOT_CellInfo_Init(nbiot_Cell10,
                                                    v_SecondaryFrequencyBand,
                                                    p_SysinfoCombination,
                                                    p_DoAttachWithoutPDN,
                                                    v_SubCarrierSpacingUL,
                                                    false/*WA#GB*/);
<<SKIPPED CODE>>


1.4 module NBIOT_CellInfo
	Template name
	module NBIOT_CellInfo

	Reason for change
	As a result of the changes 1.1 the module EUTRA_Parameters needs to be imported in this module (for px_eSecondaryFrequencyBand)

	Summary of change
	Imported module EUTRA_Parameters.ttcn into module NBIOT_CellInfo.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>

module NBIOT_CellInfo {
  import from NBIOT_RRC_ASN1_Definitions language "ASN.1:2002" all with {encode "UNALIGNED_PER_OctetAligned"};
  import from NBIOT_Imported_EUTRA_ASN1_Types all;
  import from CommonDefs all;
  import from CommonAspDefs all;
  import from NBIOT_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_NB_ASP_TypeDefs all;
  import from NBIOT_ASP_TypeDefs all; // NBIOT ASPs and port definitions
  import from NAS_CommonTypeDefs all;
  import from EPS_NAS_TypeDefs all;
  import from Parameters all;
  import from EUTRA_NB_Parameters all;
  import from NBIOT_Component all;
  import from NBIOT_Parameters all;
  import from NBIOT_SecurityFunctions all;
  import from NBIOT_SysInfo_Templates all;
  import from NBIOT_CellInfoFrequency all;
  import from EUTRA_CellInfoFrequencyRF all;
  import from EUTRA_NB_AuxiliaryFunctions all;
  import from TestcaseProperties all;
  <<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>

module NBIOT_CellInfo {
  import from NBIOT_RRC_ASN1_Definitions language "ASN.1:2002" all with {encode "UNALIGNED_PER_OctetAligned"};
  import from NBIOT_Imported_EUTRA_ASN1_Types all;
  import from CommonDefs all;
  import from CommonAspDefs all;
  import from NBIOT_CommonDefs all;
  import from EUTRA_NB_CommonDefs all;
  import from EUTRA_NB_ASP_TypeDefs all;
  import from NBIOT_ASP_TypeDefs all; // NBIOT ASPs and port definitions
  import from NAS_CommonTypeDefs all;
  import from EPS_NAS_TypeDefs all;
  import from Parameters all;
  import from EUTRA_NB_Parameters all;
  import from NBIOT_Component all;
  import from NBIOT_Parameters all;
  import from NBIOT_SecurityFunctions all;
  import from NBIOT_SysInfo_Templates all;
  import from NBIOT_CellInfoFrequency all;
  import from EUTRA_CellInfoFrequencyRF all;
  import from EUTRA_NB_AuxiliaryFunctions all;
  import from EUTRA_Parameters all; //WA#GB
  import from TestcaseProperties all;
<<SKIPPED CODE>>


